

HALLEY'S COMET - NUMERICAL CALCULATION OF ORBIT

Link to: [physicspages home page](#).

To leave a comment or report an error, please use the auxiliary blog and include the title or URL of this post in your comment.

Post date: 4 Jun 2023.

We can use Kepler's laws to generate numerical solutions for orbits on a computer. Carroll & Ostlie provide a program called *Orbit* (in Fortran or C++) that produces textual output that can be input to a spreadsheet or graphing program. However, this is a bit primitive, so I decided to implement a similar routine in Maple, since Maple has built-in plotting features.

The idea is to simulate a complete orbit of a planet by calculating its distance r from the focus of the ellipse (essentially, the distance from the Sun) as a function of the angle θ from perihelion. The equation of an ellipse is:

$$r = \frac{a(1 - e^2)}{1 + e \cos \theta} \quad (1)$$

Given the planet's period P we can calculate its semimajor axis a from Kepler's third law:

$$P^2 = \frac{4\pi^2}{GM} a^3 \quad (2)$$

so the equation 1 becomes

$$r = \left(\frac{GMP^2}{4\pi^2} \right)^{1/3} \frac{(1 - e^2)}{1 + e \cos \theta} \quad (3)$$

To use this formula in a numerical solution, we need to know how much θ changes for a given time increment dt . We can use Kepler's second law in the form

$$\frac{dA}{dt} = \frac{L}{2\mu} \quad (4)$$

which gives the rate at which area is swept out in terms of the constant total angular momentum L and reduced mass μ . The angular momentum is

$$L = \mu \sqrt{GMa(1 - e^2)} \quad (5)$$

The area increment dA is given in terms of $d\theta$ by

$$dA = \frac{r d\theta}{2\pi r} \pi r^2 = \frac{1}{2} r^2 d\theta \quad (6)$$

so we get

$$d\theta = \frac{L}{\mu r^2} dt = \frac{\sqrt{GMa(1-e^2)}}{r^2} dt \quad (7)$$

The numerical solution divides the period P up into n equal time intervals dt , and starts with r at perihelion and $\theta = 0$. We then use the value of r to calculate $d\theta$, add this to the current value of θ and calculate the next value of r from 3. We then use the new value of r to get the next $d\theta$ and so on until we've covered a complete period.

The Maple code for doing this is:

```

with(plots):
G := 0.6673e-10;
AU := 0.14959787066e12;
M_sun := 0.19891e31;
yr := 31558145.0;
rad2deg := 180/Pi;
secsYear := 365.25*(3600*24);
orbit := proc (M_strsun, a__AU, e, n)
local M_star, a, P, dt, t, theta, LoM, r, i, dtheta;
M_star := M_strsun*M_sun;
a := a__AU*AU;
P := sqrt(4*Pi^2*a^3/(G*M_star));
dt := P/(n-1);
t := Array(0 .. n, (i) -> i*dt);
r := Array(0 .. n, datatype = float);
theta := Array(0 .. n, datatype = float);
theta[0] := 0.;
LoM := sqrt(G*M_star*a*(1-e^2));
for i from 0 to n-1 do
r[i] := a*(1-e^2)/(1+e*cos(theta[i]));
dtheta := LoM*dt/r[i]^2;
theta[i+1] := evalf(theta[i]+dtheta);
if 0 < i and 1.0 < r[i]/AU and r[i-1]/AU < 1.0 then
print("Passes 1 AU at t = ", evalf(t[i]/secsYear))
end if
end do;
r[n] := a*(1-e^2)/(1+e*cos(theta[n]));
theta := theta*rad2deg;
r := r/AU;

```




FIGURE 1. Orbit of Halley’s comet. Radial distance in AU.

```
print(polarplot(r, theta, angularunit = degrees, colour=red));
print(plot(t/secsYear, r, labels = ["t (years)", "r (AU)"], colour=blue))
end proc;
```

The code pretty much just implements the algorithm given above. The Maple procedure *orbit* on line 8 takes as its arguments the mass of the central star in solar masses, the semimajor axis of the planet in AU, the eccentricity e and the number of time steps n . It then converts these quantities into SI units using the conversion factors given at the start, creates arrays for the time t , the radius r and the angle θ , calculates L/μ (as LoM), and then enters a for loop to calculate r and $d\theta$ for each time increment. The 'if' statement finds the time at which the planet crosses from $r < 1$ AU to $r > 1$ AU and prints this out.

After the loop, we calculate the final value of r , convert θ and r to degrees and AU, respectively, and print out a couple of plots. Fig. 1 is a polar plot of r as a function of θ , so it shows the elliptical orbit. Fig. 2 graphs r as a function of t (the latter in years) for one complete period.

For Halley’s comet, $P = 76$ years and $e = 0.9673$ from which we find $a = 17.943$ AU. If we input these values (along with $M_strsun = 1$ since the central star is the Sun), and choose $n = 10000$ time increments, we get the time at which the comet first crosses $r = 1$ AU after perihelion as $t = 0.1064$ years or around 39 days.




FIGURE 2. Distance of Halley's comet from the Sun, in AU.

PINGBACKS

Pingback: Elliptical orbits - numerical simulation

Pingback: Halley's comet - an application of Kepler's laws

Pingback: Oppositions of Mars - numerical calculation